
Úloha 3	Složitější OO program v Pascalu

	Tato úloha může být rozšířením úlohy 2. Jde o to vytvořit složitější objektový program, v němž budou rozumně použity virtuální metody, instance tříd budou zřizovány dynamicky a bude jich v programu více (budou uloženy ve spojovém seznamu nebo alespoň v poli). V programu budou alespoň dvě hierarchie tříd a jejich objekty budou spolu komunikovat. Veškeré funkce programu ( komunikace s uživatelem, řízení objektů apod. ) budou realizovány jako metody tříd. Zadání programu a objektový model je třeba předem odsouhlasit s vyučujícím.

	Navrhněte program, který bude fungovat jako šetřič obrazovky. Program bude zobrazovat na obrazovce ciferník analogových hodin o náhodné velikosti a barvě na náhodné pozici. Počet objektů typu ciferník nechť lze během programu dynamicky měnit. Dále nechť program zobrazuje barevný rámeček s textem a druhý s aktuálním datumem. S textovým rámečkem lze nezávisle pohybovat po obrazovce. V programu využijte hierarchii tříd, dědičnost a polymorfismus dle objektového modelu.

Objektový model:

Zdrojový program:

Výpis unity _OBJEKT.

_OBJEKT.PAS��unit _Objekt;

{Deklarace nekonkretniho objektu s vlastnostmi:Viditelnost, barva a souradnice}



Interface



Type Objekt=object

      Constructor Init (Xi,Yi:integer);

      Destructor  Done; virtual;

      Procedure   Zobraz; virtual;

      Procedure   Zhasni; virtual;

      Function    CtiX:integer;

      Function    CtiY:integer;

      Function    CtiStav:boolean;

      Procedure   NastavBarvu(B:byte);

      Function    CtiBarvu:byte;

      Procedure   Posuv (Xi,Yi:integer);

     Private

      Viditelnost:boolean;

      Barva:byte;

      X,Y:integer;

     end;





Implementation

Uses Graph;



Constructor Objekt.Init (Xi,Yi:integer);	{Zakladni naplneni dat}

 begin

  X:=Xi;

  Y:=Yi;

  Viditelnost:=false;

  NastavBarvu(Graph.GetColor);

 end;



Destructor Objekt.Done;

 begin

 end;



Procedure Objekt.Zobraz;

 begin

  Viditelnost:=True;

 end;



Procedure Objekt.Zhasni;

 begin

  Viditelnost:=False;

 end;



Function Objekt.CtiX:integer;

 begin

  CtiX:=X;

 end;



Function Objekt.CtiY:integer;

 begin

  CtiY:=Y;

 end;



Function Objekt.CtiStav:boolean;

 begin

  CtiStav:=Viditelnost;

 end;



Procedure Objekt.NastavBarvu(B:byte);

 begin

  Barva:=B;

  Graph.SetColor(B);

 end;



Function Objekt.CtiBarvu:byte;

 begin

  CtiBarvu:=Barva;

 end;



Procedure Objekt.Posuv(Xi,Yi:integer);

 begin

  X:=Xi;

  Y:=Yi;

 end;



End.

{Created by MiK}��

Výpis unity _KRUH.

_KRUH.PAS��Unit _Kruh;

{Deklarace graficke kruznice o polomeru R jako potomek tridy OBJEKT}



Interface

Uses _Objekt;



Type pKruh=^Kruh;

     Kruh=object(Objekt)

      Constructor Init(Xi,Yi,Ri:integer);

      Destructor  Done; virtual;

      Procedure   Zobraz; virtual;

      Procedure   Zhasni; virtual;

      Function    CtiR:byte;

      Procedure   NastavR (Ri:byte);

      Procedure   Posuv(Xi,Yi:integer);

      Procedure   Zvetsi(C:integer);

     Private

      R:byte;

     end;





Implementation

Uses Graph;



Constructor Kruh.Init(Xi,Yi,Ri:integer);	{Zakladni naplneni dat}

 begin

  Objekt.Init(Xi,Yi);

  R:=Ri;

 end;



Destructor Kruh.Done;

 begin

 end;



Procedure Kruh.Zobraz;

 begin

  Objekt.Zobraz;

  Graph.Circle(CtiX,CtiY,R);

 end;



Procedure Kruh.Zhasni;

var pom:byte;

 begin

  Objekt.Zhasni;

  pom:=CtiBarvu;

  NastavBarvu(Graph.GetBkColor);

  Graph.Circle(CtiX,CtiY,R);

  NastavBarvu(pom);

 end;



Function Kruh.CtiR:byte;

 begin

  CtiR:=R;

 end;



Procedure Kruh.NastavR (Ri:byte);

 begin

  R:=Ri;

 end;



Procedure Kruh.Posuv (Xi,Yi:integer);

 begin

  Zhasni;

  Objekt.Posuv(Xi,Yi);

  Zobraz;

 end;



Procedure Kruh.Zvetsi(C:integer);

 begin

  Zhasni;

  R:=R+C;

  if R<30 then R:=30;

  if R>150 then R:=150;

  Zobraz;

 end;



End.

{Created by MiK}��

Výpis unity _CIFERNI.

_CIFERNI.PAS��Unit _Ciferni;

{Deklarace ciferniku hodin jako potomek tridy KRUH}



Interface



Uses _Kruh;



Type pCifernik=^Cifernik;

     Cifernik=object(Kruh)

      Constructor Init(Xi,Yi,Ri:integer);

      Destructor  Done; virtual;

      Procedure   Zobraz; virtual;

      Procedure   Zhasni; virtual;

      Procedure   DejCas;

     Private

      h,m,s:word;

      Procedure   Spocti(cas:word;del:real;var xi,yi:integer);

      Procedure   Rucicka(col,X1,Y1:word; X2,Y2:integer);

     end;





Implementation



Uses Dos, Graph;



Const dh=0.72;					{index delky hodinove rucicky}

      dm=0.85;					{index delky minutove rucicky}

      ds=0.97;					{index delky vterinove rucicky}

      car=0.1;



Constructor Cifernik.Init(Xi,Yi,Ri:integer);	{Zakladni naplneni dat}

 begin

  Kruh.init(Xi,Yi,Ri);

  Cifernik.DejCas;

 end;



Destructor Cifernik.Done;

 begin

 end;



Procedure Cifernik.Zobraz;

var dx,dy,x,y,r:integer;

 begin

  X:=CtiX;

  Y:=CtiY;

  R:=CtiR;

  Kruh.Zobraz;

  Graph.Circle(X,Y,R+3);			{Dvojita kruznice}

  Rectangle(X+1,Y-R,X-1,Y-R+round(Car*R));

  Rectangle(X+R,Y+1,X+R-round(Car*R),Y-1);

  Rectangle(X+1,Y+R,X-1,Y+R-round(Car*R));

  Rectangle(X-R,Y+1,X-R+round(Car*R),Y-1);



  Spocti(h,R*dh,dX,dY);				{Vypocet a vykresleni rucicek}

   SetLineStyle(SolidLn,0,ThickWidth);

   Rucicka(Green,X,Y,X+dX,Y-dY);

  Spocti(m,R*dm,dX,dY);

   SetLineStyle(SolidLn,0,NormWidth);

   Rucicka(LightBlue,X,Y,X+dX,Y-dY);

  Spocti(s,R*ds,dX,dY);

   SetLineStyle(DashedLn,0,NormWidth);

   Rucicka(LightRed,X,Y,X+dX,Y-dY);

  SetLineStyle(SolidLn,0,NormWidth);

 end;



Procedure Cifernik.Zhasni;

var dx,dy,x,y:integer;

    pom,r:byte;

 begin

  X:=CtiX;

  Y:=CtiY;

  R:=CtiR;

  Kruh.Zhasni;

  Pom:=CtiBarvu;

  NastavBarvu(Graph.GetBkColor);

  Graph.Circle(X,Y,R+3);

  Rectangle(X+1,Y-R,X-1,Y-R+round(Car*R));

  Rectangle(X+R,Y+1,X+R-round(Car*R),Y-1);

  Rectangle(X+1,Y+R,X-1,Y+R-round(Car*R));

  Rectangle(X-R,Y+1,X-R+round(Car*R),Y-1);



  Spocti(h,R*dh,dX,dY);

   SetLineStyle(SolidLn,0,ThickWidth);

   Rucicka(CtiBarvu,X,Y,X+dX,Y-dY);

  Spocti(m,R*dm,dX,dY);

   SetLineStyle(SolidLn,0,NormWidth);

   Rucicka(CtiBarvu,X,Y,X+dX,Y-dY);

  Spocti(s,R*ds,dX,dY);

   SetLineStyle(DashedLn,0,NormWidth);

   Rucicka(CtiBarvu,X,Y,X+dX,Y-dY);

  SetLineStyle(SolidLn,0,NormWidth);

  NastavBarvu(pom);

 end;



Procedure Cifernik.DejCas;			{Zjisteni systemoveho casu}

 var ss:word;

 begin

  Dos.Gettime(H,M,S,ss);

  if h>=12 then h:=h-12;			{Uprava prom. hodina}

  h:=h*5;

  h:=h+m div 12;

 end;



Procedure Cifernik.Spocti(cas:word;del:real;var xi,yi:integer);

const K=6.28318/60;

 begin

  xi:=round(del*sin(cas*k));			{Vypocet druhe souradnice}

  yi:=round(del*cos(cas*k));			{Predavaji se nadrizene proc.}

 end;



Procedure Cifernik.Rucicka(col,X1,Y1:word; X2,Y2:integer);

 begin

  setcolor(col);				{Vykresleni rucky danou barvou}

  line(X1,Y1,X2,Y2);				{na dane pozici}

 end;



End.  

{Created by MiK}��

Výpis unity _OBDELNI.

_OBDELNI.PAS��Unit _Obdelni;

{Deklarace grafickeho obdelniku o rozmerech dX*dY jako potomek tridy OBJEKT}



Interface

Uses _Objekt;



Type pObdelnik=^Obdelnik;

     Obdelnik=object(Objekt)

      Constructor Init(Xi,Yi,dXi,dYi:integer);

      Destructor  Done; virtual;

      Procedure   Zobraz; virtual;

      Procedure   Zhasni; virtual;

      Function    CtidX:integer;

      Function    CtidY:integer;

      Procedure   Velikost (dXi,dYi:integer);

      Procedure   Posuv(Xi,Yi:integer);

      Procedure   Zvetsi(dXi,dYi:integer);

     Private

      dX,dY:integer;

     end;





Implementation

Uses Graph;



Constructor Obdelnik.Init(Xi,Yi,dXi,dYi:integer);

 begin

  dX:=dXi;

  dY:=dYi;

  Objekt.Init(Xi,Yi);

 end;



Destructor Obdelnik.Done;

 begin

 end;



Procedure Obdelnik.Zobraz;			{Objekt je urcen souradnicemi}

 begin						{leveho horniho rohu, }

  Objekt.Zobraz;				{vyskou a delkou}

  Graph.Rectangle(CtiX,CtiY,CtiX+dX,CtiY+dY);

 end;



Procedure Obdelnik.Zhasni;

var pom:byte;

 begin

  Objekt.Zobraz;

  pom:=CtiBarvu;

  NastavBarvu(Graph.GetBkColor);

  Graph.Rectangle(CtiX,CtiY,CtiX+dX,CtiY+dY);

  NastavBarvu(pom);

 end;



Function Obdelnik.CtidX:integer;

 begin

  CtidX:=dX;

 end;



Function Obdelnik.CtidY:integer;

 begin

  CtidY:=dY;

 end;



Procedure Obdelnik.Velikost(dXi,dYi:integer);

 begin

  dX:=dXi;

  dY:=dYi;

 end;



Procedure Obdelnik.Posuv (Xi,Yi:integer);

 begin

  Zhasni;

  Objekt.Posuv(Xi,Yi);

  Zobraz;

 end;



Procedure Obdelnik.Zvetsi(dXi,dYi:integer);

 begin

  Zhasni;

  dX:=dX+dXi;

  dY:=dY+dYi;

  Zobraz;

 end;



End.

{Created by MiK}��

Výpis unity _NAPIS.

_NAPIS.PAS��Unit _Napis;

{Deklarace obdelniku s textem}



Interface

Uses _Obdelni;



Type pNapis=^Napis;

     Napis=object(Obdelnik)

      Constructor Init(Xi,Yi,dXi,dYi:integer;Stri:string);

      Destructor  Done; virtual;

      Procedure   Zobraz; virtual;

      Procedure   Zhasni; virtual;

      Function    CtiBarvuT:Byte;

      Procedure   NastavBarvuT(B:byte);

      Function    DejDelku:Word;

      Function    Dejvysku:Word;

     Private

      Str:string;				{vypisovany retezec}

      BarvaT:Byte;				{barva textu}

      Delka,Vyska:word;				{delka a vyska obdelniku}

     end;





Implementation

Uses Graph;



Const HMezera=3;

      VMezera=3;



Constructor Napis.Init(Xi,Yi,dXi,dYi:integer;Stri:string);

var dXii,dYii:integer;

 begin

  NastavBarvuT(Graph.GetColor);

  Str:=Stri;

  dXii:=dXi;

  dYii:=dYi;

  if dXi<TextWidth(Str)+HMezera*2 then dXii:=TextWidth(Str)+HMezera*2;

  if dYi<TextHeight(Str)+VMezera*2 then dYii:=TextHeight(Str)+VMezera*2;

  Obdelnik.Init(Xi,Yi,dXii,dYii);

  Delka:=dXii;

  Vyska:=dYii;

 end;



Destructor Napis.Done;

 begin

 end;



Procedure Napis.Zobraz;

 begin

  Graph.SetColor(CtiBarvu);

  Obdelnik.Zobraz;				{Nejprve so zobrazi obdelnik}

  Graph.SetColor(BarvaT);			{pote text s BarvouTextu}

  Graph.OuttextXY (CtiX+HMezera,CtiY+VMezera,Str);

  Graph.SetColor(CtiBarvu);

 end;



Procedure Napis.Zhasni;

var pom:byte;

 begin

  Obdelnik.Zhasni;

  pom:=CtiBarvu;

  NastavBarvu(Graph.GetBkColor);

  Graph.OuttextXY (CtiX+HMezera,CtiY+VMezera,Str);

  NastavBarvu(pom);

 end;



Function Napis.CtiBarvuT:byte;

 begin

  CtiBarvuT:=BarvaT;

 end;



Procedure Napis.NastavBarvuT(B:byte);

 begin

  BarvaT:=B;

  Graph.SetColor(B);

 end;



Function Napis.DejDelku:Word;

 begin

  DejDelku:=Delka;

 end;



Function Napis.DejVysku:Word;

 begin

  DejVysku:=Vyska;

 end;



End.

{Created by MiK}��

Výpis unity _DATUM.

_DATUM.PAS��Unit _Datum;

{Deklarace obdelniku s datem jako spec. pripad zdedeny z tridy NAPIS}



Interface

Uses _Napis;



Type pDatum=^Datum;

     Datum=object(Napis)

      Constructor Init(Xi,Yi,dXi,dYi:integer);

      Destructor  Done; virtual;

     Private

      Dat:string;				{vypisovany retezec s datem}

     end;





Implementation

Uses Dos;



Constructor Datum.Init(Xi,Yi,dXi,dYi:integer);

var ds,ms,rs:string;

    D,M,R,Dot:word;

 begin

  Dos.GetDate(R,M,D,Dot);			{Ziskani systemoveho data}

  Str(D,ds); Str(M,ms); Str(R,rs);		{Jeho prevedeni na retezce}

  Dat:=(Ds+'.'+Ms+'.'+Rs);			{Spojeni dilcich retezcu}

  Napis.Init(Xi,Yi,dXi,dYi,Dat);

 end;



Destructor Datum.Done;

 begin

 end;



End.

{Created by MiK}��

Výpis programu.

HODINY3.PAS��Program Hodiny3_OOP_(Dynamicke);



Uses Graph,Crt,_Ciferni,_Napis,_Datum;



Const Copyright='Copyright (c) 1999, Michal Kec.';

      TXT='Toto je ScreenSaver '+#2;		{Heslo, ktere se vypisuje}

      Pocet=10;						{Omezeni poctu ciferniku}



Type Saver=object

      Procedure Init(HesloDne:string);

      Procedure Run;

      Procedure Done;

     Private

      PocCif:byte;					{aktualni pocet ciferniku}

      Hodiny:array[1..Pocet] of pCifernik;	{pole objektu typu CIFERNIK}

      Text:pNapis;					{objekt typu NAPIS}

      Dat:pDatum;						{objekt typu DATUM}

     end;





Procedure Saver.Init(HesloDne:string);

Var grDriver, grMode, ErrCode: Integer;

    s:string;

 Begin

  GetDir(0,S);					{pokousi se zjistit, kde bezi}

  if s[1]='C' then s:='C:\TP\BGI'		{predpoklada lokalni rezim}

	      else s:='Y:\BP\BGI';		{predpoklada existenci site}

  grDriver:=Detect;				{inicializace grafiky}

  InitGraph(grDriver,grMode,s);

  ErrCode:=GraphResult;

  if ErrCode <> grOk then			{selhani grafiky}

   begin

    Writeln;

    WriteLn('Chyba grafiky: ',GraphErrorMsg(ErrCode));

    delay(1000);

    halt(5);					{abnormalni ukonceni programu}

   end;

  Randomize;

					{vytvoreni 3 dynamickych objektu}

  New(Text,Init(Random(GetmaxX-Text^.DejDelku),Random(GetmaxY-Text^.DejVysku),0,0,HesloDne));

  New(Dat,Init(Random(GetmaxX-Dat^.DejDelku),Random(GetmaxY-Dat^.DejVysku),0,0));

  New(Hodiny[1],Init(320,200,50));

  PocCif:=1;					{mame prave jedny hodiny}

 end;



Procedure Saver.Run;

Var k:byte;

    i:shortint;					{index cyklu FOR}

    ch:char;

 begin

  Repeat;						{HLAVNI CYKLUS}

   Dat^.NastavBarvu(Random(15)+1);

   Repeat Dat^.NastavBarvuT(Random(15)+1) until Dat^.Ctibarvu<>Dat^.CtiBarvuT;

   Dat^.Posuv(Random(GetmaxX-Dat^.DejDelku),Random(GetmaxY-Dat^.DejVysku));

   Text^.Posuv(Random(GetmaxX-Text^.DejDelku),Random(GetmaxY-Text^.DejVysku));

   Hodiny[PocCif]^.Posuv(random(340)+150,random(180)+150);

   Hodiny[PocCif]^.Zvetsi(Random(50)-25);



   for i:=10 downto -10 do			{SMYCKA HODIN}

    if not keypressed then			{nebyla-li stisknuta klavesa}

     begin						{provadi se prekresleni hodin}

      Text^.NastavBarvu(Random(15)+1);

      Text^.NastavBarvuT(Random(15)+1);

      Text^.Zobraz;

      Dat^.Zobraz;

      Hodiny[PocCif]^.NastavBarvu(random(15)+1);

      for k:=1 to PocCif do			{prekresleni vsech HODIN}

       begin

	Hodiny[k]^.Zhasni;

	Hodiny[k]^.Dejcas;

	Hodiny[k]^.Zvetsi(i+1 div (abs(i)+1));

       end;

      delay(200 div PocCif);			{zpozdeni}

     end

     else						{byla-li stisknuta klavesa}

     begin

      inc(i);

      ch:=readkey;				{testujeme udalost}

      if ch=#0 then ch:=readkey;

      case ch of

       '+':begin					{pridani objektu CIFERNIK}

	    if PocCif<Pocet then

	     begin

	      Hodiny[PocCif]^.NastavBarvu(PocCif);

	      inc(PocCif);			{dalsi prvek pole}

	      New(Hodiny[PocCif],Init(random(340)+150,random(180)+150,Random(25)+25));

	     end

	     else					{mame-li MAX prvku}

	     begin					{nemuzeme zadny pridat}

	      Sound(5000);

	      Delay(1);

	      Nosound;

	     end;

	   end;

       '-':begin					{odebrani objektu CIFERNIK}

	    if PocCif>1 then

	     begin

	      hodiny[PocCif]^.Zhasni;

	      Dispose(Hodiny[PocCif],Done);	{uvolneni pameti}

	      dec(PocCif);

	     end

	     else					{mame-li jen jeden prvek}

	     begin					{nemuzeme ho odebrat}

	      Sound(5000);

	      Delay(1);

	      Nosound;

	     end;

	   end;

       #72:Text^.Posuv(Text^.CtiX,Text^.CtiY-5);	{posuny objektu TEXT- nahoru}

       #80:Text^.Posuv(Text^.CtiX,Text^.CtiY+5);			   {- dolu}

       #75:Text^.Posuv(Text^.CtiX-5,Text^.CtiY);			   {- doleva}

       #77:Text^.Posuv(Text^.CtiX+5,Text^.CtiY);			   {- doprava}

       #71:Text^.Posuv(0,0);							   {- home}

       #73:Text^.Posuv(GetmaxX-text^.dejDelku,0);			   {- PgUp}

       #79:Text^.Posuv(0,GetmaxY-text^.DejVysku);			   {- Delete}

       #81:Text^.Posuv(GetmaxX-Text^.DejDelku,GetmaxY-Text^.DejVysku); {PgDown}

       '`':;

       else 					{byla-li zmacknuta jina klavesa}

       exit;

      end;

     end;						{KONEC SMYCKY HODIN}

   while keypressed do ch:=readkey;

  until keypressed;				{KONEC HLAVNIHO CYKLU}

 end;



Procedure Saver.Done;

Var k:byte;

 begin

  Dat^.Zhasni;					{zhasneme DATUM}

  Text^.Zhasni;					{zhasneme TEXT}

  for k:=PocCif downto 1 do			{zhasneme a uvolnime CIFERNIKy}

   begin

    Hodiny[k]^.Zhasni;

    Dispose(Hodiny[k],Done);

   end;

  Dispose(Dat,Done);				{uvolnime DATUM}

  Dispose(Text,Done);				{uvolnime TEXT}

  CloseGraph;					{uzavreme graficky rezim}

  Writeln;

  Writeln(Copyright);				{nakonec info :-)}

  Exit;

 end;





{------------------------Vlastni telo programu-----------------------------}

Var Prog:Saver;



Begin

 Prog.Init(TXT);

 Prog.Run;

 Prog.Done;

End.

{Created by MiK}



Potrebuje tyto unity:	_Objekt.tpu

			_Kruh.tpu

			_Ciferni.tpu

			_Obdelni.tpu

			_Napis.tpu

			_Datum.tpu��


